include("${CMAKE_CURRENT_LIST_DIR}/librw-targets.cmake")

set(LIBRW_PLATFORM "@LIBRW_PLATFORM@")
set(LIBRW_PLATFORMS "@LIBRW_PLATFORMS@")
set(LIBRW_PLATFORM_@LIBRW_PLATFORM@ ON)

if(LIBRW_PLATFORM_GL3)
    set(LIBRW_GL3_GFXLIB "@LIBRW_GL3_GFXLIB@")
    set(LIBRW_GL3_GFXLIBS "@LIBRW_GL3_GFXLIBS@")

    set(OpenGL_GL_PREFERENCE GLVND)
    find_package(OpenGL)
    if(NOT TARGET OpenGL::OpenGL AND NOT TARGET OpenGL::EGL AND NOT TARGET OpenGL::GL)
        message(FATAL_ERROR "find_package(OpenGL) failed: no target was created")
    endif()

    if(LIBRW_GL3_GFXLIB STREQUAL "GLFW")
        if (NOT TARGET glfw)
            find_package(glfw3 REQUIRED)
        endif()
    elseif(LIBRW_GL3_GFXLIB STREQUAL "SDL2")
        if (NOT TARGET SDL2::SDL2)
            find_package(SDL2 CONFIG)
            if (NOT TARGET SDL2::SDL2)
                find_package(SDL2 MODULE REQUIRED)
            endif()
        endif()
    endif()
endif()
